

Importing User Accounts in a CSV File using Visual Basic Script

You can use TRCLI.exe (TRCLILT.exe for LT Edition) and a Visual Basic Script to import user accounts in a CSV text file. Assume that a CSV text file named Users.csv contains following lines;

```
User-1, Pass-1
User-2, Pass-2
:
User-n, Pass-n
```

You can use following Visual Basic Script to import user accounts listed in Users.csv;

```
Dim username, groupid, csvValues, strArgs

Set strArgs = WScript.Arguments

If strArgs.length < 1 then
    WScript.Echo "Usage : TR-CSV-Import [CSV File] [GroupID]"
    WScript.Quit
End If

If strArgs.length < 2 then
    groupid = "Default"
Else
    groupid = Trim(strArgs(1))
End If

Set WshShell = WScript.CreateObject ("WScript.Shell")
Set objFSO = CreateObject("Scripting.FileSystemObject")
Set objFile = objFSO.OpenTextFile(objFSO.GetAbsolutePathName("") & "\" & Trim(strArgs(0)))
Set objRegExp = CreateObject("VBScript.RegExp")

objRegExp.Pattern = "^(.+),(.+)$"

Do Until objFile.AtEndOfStream
    strLine = objFile.ReadLine
    Set colMatches = objRegExp.Execute(strLine)

    If colMatches.Count > 0 Then
        csvValues = Split(strLine, ",")
        username = Trim(csvValues(0))

        ' Following line runs trcli.exe and adds a user profile. Use trclilt for LT edition.

        strRun = "%comspec% /c trcli -u " & username & " " & Trim(csvValues(1)) & " " & groupid
        WshShell.Run strRun, 0, True

        ' You can add additional attributes to user profiles.
        ' Following lines add Service-Type = Framed (Check)

        strRun = "%comspec% /c trcli -a " & username & " ""ietf|6"" 2 check"
        WshShell.Run strRun, 0, True
    End If
Loop

objFile.Close
```

You can customize this script to import more attributes to user profiles. Script accepts two parameters. First parameter is the name of the CSV file and the second parameter is Group ID of the user profile to be imported.